AD-A079 226 ARMY PERSONNEL RESEARCH OFFICE WASHINGTON DC F/G 15/5

A LINEAR PROGRAM FOR TRANSPORTATION COST MINIMIZATION OF BCT-AI--ETC(U)

APR 63 E P STICHMAN

UNCLASSIFIED APRO-RM-63-7 NL

1 OF 1

AD-
A079226

END
DATE
FILMED
2-80
DDC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

ADA079226

79 12 18 187

Army Project Number
0J95-60-001

Future Combat c-00

9 Research Memorandum 63-7

6 A LINEAR PROGRAM FOR TRANSPORTATION COST MINIMIZATION OF BCT-AIT SHIPPING.

10 Eugene P. Stichman

Robert F. Boldt, Task Leader

12 21

14 APRO-RM-63-7

Submitted by

Philip J. Bersh
Chief, Combat Systems Research Laboratory

11 April 1963

040 650

J03

A LINEAR PROGRAM FOR TRANSPORTATION COST MINIMIZATION OF BCT-AIT SHIPPING

BACKGROUND

The Combat Allocation Task was initiated by USAPRO in 1957 in response to USCONARC's concern that an insufficient supply of able personnel was reaching combat units. The objective was to provide a system of assigning enlisted men to advanced training resulting in an adequate and equitable distribution of available aptitudes among combat units, combat support units, and other branches of the Army. The system chosen must provide optimal allocation of personnel to MOS based on Army manpower requirements and personnel qualifications. In anticipation of Army needs, USAPRO personnel recognized that an optimal system for allocating personnel would require the use of stored-program computing equipment. During FY 1961 and FY 1962, the Combat Allocation Task began the development, elaboration, and testing of a computer-based assignment method for allocation of uncommitted enlisted personnel to MOS training.

In September, 1962, the Adjutant General's Office formulated plans for an Automatic Data Processing System (ADP) for handling the flow of trainees through the training base. Under this system, processing is divided into four phases, or ACT's (Automated Control of Trainees). ACT I is concerned with initial input and control of a master file. ACT II treats the distribution of basic combat training (BCT) personnel to advanced individual training (AIT) and/or directly to unit assignment, while ACT III handles the assignment from AIT to the unit. ACT IV is the statistical analysis program for the total ADP system package. Under this proposed ADP system, Army personnel processing activities were unified under TAG control.

As an outgrowth of this unification, a joint TAG-USAPRO effort was initiated and directed primarily toward the development of automated procedures for providing trainees in BCT with their next assignment (ACT II). USAPRO expanded its earlier effort to include the development and testing of a computer program (Transportation Program) for the minimization of costs related to transportation of allocated personnel from BCT to AIT for their respective MOS training. This new effort was in addition to the continued testing of the previously developed computer program for the optimal allocation of uncommitted trainees. The Transportation Program for shipping cost minimization includes as input both the assignments to MOS of uncommitted enlisted personnel as allocated by the optimal allocation program and the assignments of committed, earmarked, and special category enlisted personnel as determined by TAG-designed methods. The present Research Memorandum describes the development of the Transportation Program.

THE TRANSPORTATION PROBLEM

The problem of transportation costs arises when BCT trainees at more than one center are assigned to advanced training in an MOS which is offered at more than one location. The cost of travel from BCT to AIT is determined by the mileage between the two locations. Clearly, certain routes are preferable to certain others on a sheer dollar/mile basis. However, the number of trainees who can make a preferred trip is limited by the number of trainees at a given BCT center and the capacity of the advanced training location. The optimum solution to this problem is to ship all trainees from BCT to AIT in such a way that the total cost of the transportation is at a minimum.

Problems of the kind mentioned above are formally stated as follows:

Let c_{ij}	be the unit cost of shipping a commodity from location i to destination j.
x_{ij}	be the number of units of the commodity to be shipped from location i to destination j.
$z = \sum_i \sum_j x_{ij} c_{ij}$	be the total cost of shipping incurred for a given set of x_{ij}.
P_i	be the amount to be shipped from location i.
Q_j	be the amount needed at destination j.

(The problem has the restraint for solution that $\sum_i P_i = \sum_j Q_j$.)

Then, minimize the value of z subject to the linear restraints:

$$\sum_j x_{ij} = P_i$$

$$\sum_i x_{ij} = Q_j$$

$$x_{ij} \geq 0$$

Various solutions to the problem stated above, known as the Transportation Problem, have been available. Although the arithmetic required is long and tedious, only a mechanical repetition of a sequence of precisely defined steps is involved. The time required can be reduced to minutes by the use of high speed computers where each operation is performed in micro-seconds and the complete sequence is automatic.

The first step in the solution of the transportation problem is to develop a feasible solution, one which satisfies the movement requirements. The traditional technique involves a routine of iterations through a succession of feasible solutions of gradually diminishing cost to a feasible solution of minimal cost (the optimal solution). This procedure is lengthy and inefficient for computer operation. Therefore, a newer method--the Hungarian Method--was adopted.1/

In using the Hungarian Method for solution of the transportation problem, one begins with an initial optimal, albeit infeasible, solution to the problem (in contrast to the traditional technique mentioned above) and iterates through other optimal solutions until some feasible solution is reached. A feasible solution means a set of individual allocations such that all restraints are satisfied, that is, existing surpluses are removed and existing deficiencies are filled.

GENERALIZED COMPUTER PROGRAM

The program, while developed to minimize transportation costs for BCT-AIT shipping, is adaptable to related problems. The general format of the program will accommodate four (4) character cell entries with the initial COST2/ matrix (c_{ij} matrix), the final transportation matrix (x_{ij} matrix)--called SEND--and the ROW (P_i) and COL (column) (Q_j) restraints. The row and column entries, therefore, may not exceed a value of 9,999. An additional restriction is placed on the maximum entries for the cost matrix. Due to the arithmetic operations of the computer and the numerical aids required for programming facilitation, no entry in the cost matrix may exceed a value of 3,499. As an alternate restriction, the sum of the two largest entries combined may not exceed a total value of 6,999. Either of these two parameter limitations must be observed unless special conditions pertain to the specific problem.

1/ The basic steps of this solution were obtained during attendance by USAPRO personnel at the University of Michigan's Summer Conferences on System Engineering, 1961. This method is presented in Appendix A.

2/ The use of capital letters here refers to actual abbreviations and mnemonic symbols used in the program.

VARIABLES REQUIRING DEFINITION

To operate within the framework of the program, certain variables require specific definition for each individual problem:

Variables	Size
SEND	Number of cell entries x 4
COST	Number of cell entries x 4
COL	Number of columns x 4
ROW	Number of rows x 4
M	Number of rows
N	Number of columns
FOURM	Number of rows x 4
MN	Number of rows x number of columns

The term SEND refers to the computer area set aside for compilation of the final transportation matrix, which contains the solution to the problem. Each cell entry yields the number of units to be shipped to the destination specified by the column from the origin specified by the row. COST refers to the area set aside for the initial input to the cost matrix (the cell entries). Each entry is the cost, or scaled cost, of shipping one unit from a given origin to a given destination. COL is the area set aside for input of the capacity quotas at each destination. This value determines the number to be shipped to this destination. ROW is the area set aside for input of the units available for transport at each origin. The other terms must be specified for each specific problem.

COMPUTER PROGRAM FOR BCT-AIT TRANSPORTATION

Given appropriate ROW, COL, and COST inputs, the program results in a solution such that the distribution of numbers of trainees from multiple BCT centers to multiple AIT locations is accomplished with the total cost of the necessary differential transportation at a minimum. In minimizing the cost, both the manpower available at the BCT centers and the quotas for training at the AIT locations are taken into account. Because only 13 percent of the AIT courses have multiple locations, this program operates efficiently, treating each multiple destination MOS separately.

The major restriction is that the total available personnel, the sum of the row values, must equal the total assignment quotas at the destinations, that is, the sum of the column values $(\sum_i P_i = \sum_j Q_j)$. Any discrepancy between these two sums will result in an invalid solution if any solution is obtained.

PROGRAM SYMBOLOGY

The first step in the design of the actual program was to chart the flow of the Hungarian Method in computer language. This crucial initial effort was accomplished by Dr. Marjorie O. Chandler of the USAPRO research staff, who utilized FORTRAN symbology for this chart. When FORTRAN was found to be inefficient for use with the IBM 1401 system, the logical FORTRAN operations were translated into AUTOCODER (a symbology more compatible with 1401 systems), and the program was written in this symbolic language. AUTOCODER, one of two basic symbolic programming aids for the 1401 Data Processing Systems, consists of a set of language specifications used to write the source program. A processor program translates the symbolic language program into the actual machine language program. AUTOCODER permits the user to define areas and write instructions by writing symbolic statements. These statements are written using mnemonic operation codes and the symbolic names with which the user defines his fields.

REQUIRED DATA

The data necessary for the operation of this program are the initial cost matrix, the column values, and the row values. These are punched on standard IBM punch cards, using four columns for each entry. All 80 columns may be used if necessary. The first card is for the cost matrix.3/ If the matrix is larger than can be accommodated on one card, additional cards may be used as needed. The second card, or group of cards, specifies the column values. The third specifies the row values. These data cards should always be introduced following the object program (actual machine language program) in the order in which they have been described.

SAMPLE SOLUTION

The example used for program presentation comprises a 7 x 8 cost matrix. Computer running time for this sample problem using actual cost data was approximately 30 seconds from the time the first card of the object program was read until the solution was output on the printer. This same problem, when solved manually, required approximately two hours to reach the same solution.

The complete program is presented as Appendix B.

3/ Data must be entered onto cards by elements within columns, starting with column 1 of the matrix. The first entry comprises card columns 1-4.

REFERENCES

Churchman, C. W., Ackoff, R. L., and Arnoff, E. L. Introduction to operations research. New York: John Wiley and Sons, Inc. 1957.

IBM Reference Manual, IBM 1401 data processing system. Endicott, N. Y. 1962.

Sasieni, M., Yaspan, A., and Friedman, L. Operations research -- methods and problems. New York: John Wiley and Sons, Inc. 1959.

APPENDIX A

TRANSPORTATION PROBLEM BY THE HUNGARIAN METHOD

APPENDIX A

TRANSPORTATION PROBLEM BY THE HUNGARIAN METHOD

1. Subtract from each row its smallest element. Then subtract from each column of the resulting matrix its smallest element.

2. Pick a trial set of quotas by assigning them to zeros, subtracting appropriately from the discrepancies.

3. Cover each column whose discrepancy is zero.

4. Find an uncovered zero (if there is none, go to step 7). Prime it. Check the discrepancy of the row; if it is not zero, go to step 6. If it is zero, go to step 5.

5. Cover the row, and for each twice covered essential zero, star the zero and uncover its column. Go to step 4.

6. There now exists a unique chain, starting at the primed zero, going vertically to a starred zero, horizontally to a primed zero, etc., and ending on a primed zero (with no starred zero in the column).

   Find the smallest of the following numbers; the discrepancy of the row of the first primed zero in the chain; the discrepancy of the column of the last primed zero in the chain; the quota of each starred zero in the chain. This number is to be subtracted from each of these two discrepancies, and from the quota of every starred zero in the chain, and to be added to the quota of every primed zero in the chain. Now erase all primes and stars, uncover all rows, and, if the solution is not complete, go to step 3.

7. Find the smallest uncovered element in the matrix. Add this element to the covered rows and subtract it from the uncovered columns (or add it to the covered columns and subtract it from the uncovered rows) (or add it to the twice-covered elements and subtract if from the uncovered elements). Do not change any stars, primes, or coverings. Go to step 4.

NOTE: "Discrepancies" are amounts to be shipped which have not yet been assigned. "Quotas" are amounts which have already been assigned to particular elements of the matrix (i.e., particular routes). An essential zero is one whose quota is greater than zero.

APPENDIX B

AUTOCODER PROGRAM FOR BCT-AIT TRANSPORTATION

```
                CTL     551
        SEND    DA      56X4,X3
        COST    DA      56X4,X3
        COL     DA      8X4,X2
        ROW     DA      7X4,X1
        X1      EQU     89
        X2      EQU     94
        X3      EQU     99
        SET     DCW     0
        M       DCW     &0007
        N       DCW     &0008
        FOURM   DCW     028
        ONE     DCW     &0001
        FOUR    DCW     004
        ZERO3   DCW     000
        ZERO4   DCW     &0000
        K       DCW     &9999
        INDEX   DCW     &0000
        MIN     DCW     &0000
        W       DCW     003
        MONE    DCW     -0001
        MI      DCW     &0000
        NJ      DCW     &0000
        TS1     DCW     000
        TS2     DCW     000
        TS3     DCW     000
        TS4     DCW     &0000
        TS5     DCW     &0000
        NINT    DCW     &9000
        NINTO   DCW     &9001
        AYTT    DCW     &8000
        AYTTO   DCW     &8001
        MN      DCW     &0056
        OBJECT  MLC     MN,NJ
                CS      99
                SW      97
                SW      92
                SW      87
                SW      1
                R
                MLCWA   80,COST&79
                R
                MLCWA   80,COST&159
                R
                MLCWA   64,COST&223
                R
                MLCWA   32,COL&31
                R
                MLCWA   28,ROW&27
                MLC     ZERO3,X3
        FIRST   SW      COST&X3
                MA      FOUR,X3
                S       ONE,NJ
```

```
                C      NJ,ZERO4
                BL     FIRST
                MLC    ZERO3,X2
                MLC    N,NJ
       SECOND   SW     COL&X2
                MA     FOUR,X2
                S      ONE,NJ
                C      NJ,ZERO4
                BL     SECOND
                MLC    ZERO3,X1
                MLC    M,MI
       THIRD    SW     ROW&X1
                MA     FOUR,X1
                S      ONE,MI
                C      MI,ZERO4
                BL     THIRD
                H
                MLC    W,X1
                MLC    W,X2
                MLC    W,X3
                MLC    N,NJ
       AA       MLC    M,MI
       AB       C      ZERO4,COST&X3
                MLC    ZERO4,SEND&X3
                BU     AD
                C      ROW&X1,COL&X2
                BL     AC
                MLC    ROW&X1,SEND&X3
                S      ROW&X1,COL&X2
                MLC    ZERO4,ROW&X1
                B      AD
       AC       MLC    COL&X2,SEND&X3
                S      COL&X2,ROW&X1
                MLC    ZERO4,COL&X2
       AD       MA     FOUR,X1
                MA     FOUR,X3
                S      ONE,MI
                C      MI,ZERO4
                BL     AB
                MA     FOUR,X2
                MLC    W,X1
                S      ONE,NJ
                C      NJ,ZERO4
                BL     AA
       AE       MLC    W,X2
                MLC    N,NJ
       AF       C      ZERO4,COL&X2
                BU     AG
                MLC    MONE,COL&X2
       AG       MA     FOUR,X2
                S      ONE,NJ
                C      NJ,ZERO4
                BL     AF
```

```
AH      MLC     W,X1
        MLC     W,X2
        MLC     W,X3
        MLC     N,NJ
        MLC     M,MI
AJ      BWZ     AM,COL&X2,K
AK      C       ZERO4,COST&X3
        BU      AL
        BWZ     AL,ROW&X1,K
        B       AP
AL      MA      FOUR,X1
        MA      FOUR,X3
        S       ONE,MI
        C       MI,ZERO4
        BL      AK
        MLC     W,X1
        MLC     M,MI
        MA      FOUR,X2
        B       AN
AM      MA      FOUR,X2
        MA      FOURM,X3
AN      S       ONE,NJ
        C       NJ,ZERO4
        BL      AJ
        C       ZERO4,INDEX
        BE      CA
        MLC     ZERO4,INDEX
        B       AH
AP      A       NINT,COST&X3
        MLC     X1,TS1
        MLC     X2,TS2
        MLC     X3,TS3
        MLC     MI,TS4
        MLC     NJ,TS5
        BWZ     APP,ROW&X1,K
        C       ZERO4,ROW&X1
        BH      BA
APP     MLC     MONE,ROW&X1
        MLC     X1,X3
        MLC     W,X2
        MLC     N,NJ
AQ      C       ZERO4,SEND&X3
        BE      AR
        BWZ     AQQ,COL&X2,K
        B       AR
AQQ     A       AYTT,COST&X3
        MLC     ZERO4,COL&X2
        MLC     ONE,INDEX
AR      MA      FOUR,X2
        MA      FOURM,X3
        S       ONE,NJ
        C       NJ,ZERO4
        BL      AQ
```

```
                MLC     TS1,X1
                MLC     TS2,X2
                MLC     TS3,X3
                MLC     TS4,MI
                MLC     TS5,NJ
                B       AL
        BA      MLC     ROW&X1,MIN
        BB      BWZ     BBB,COL&X2,K
                C       ZERO4,COL&X2
                BH      BG
        BBB     S       W,X1
                MN      X1,SET
                MLC     SET,X1
                S       X1,X3
                MN      X3,SET
                MLC     SET,X3
                MLC     W,X1
                MLC     M,MI
        BC      S       AYTT,COST&X3
                C       ZERO4,COST&X3
                A       AYTTO,COST&X3
                S       ONE,COST&X3
                BE      BD
                MA      FOUR,X1
                MA      FOUR,X3
                S       ONE,MI
                B       BC
        BD      C       SEND&X3,MIN
                BL      BE
                MLC     SEND&X3,MIN
        BE      MLC     X1,X3
                MLC     W,X2
                MLC     N,NJ
        BF      S       NINT,COST&X3
                C       ZERO4,COST&X3
                A       NINTO,COST&X3
                S       ONE,COST&X3
                BE      BB
                MA      FOUR,X2
                MA      FOURM,X3
                S       ONE,NJ
                B       BF
        BG      C       COL&X2,MIN
                BL      BH
                MLC     COL&X2,MIN
        BH      MLC     TS1,X1
                MLC     TS2,X2
                MLC     TS3,X3
                MLC     TS4,MI
                MLC     TS5,NJ
                S       MIN,ROW&X1
        BJ      A       MIN,SEND&X3
                BWZ     BJJ,COL&X2,K
```

```
        C     ZERO4,COL&X2
        BH    BN
BJJ     S     W,X1
        MN    X1,SET
        MLC   SET,X1
        S     X1,X3
        MN    X3,SET
        MLC   SET,X3
        MLC   W,X1
        MLC   M,MI
BK      S     AYTT,COST&X3
        C     ZERO4,COST&X3
        A     AYTTO,COST&X3
        S     ONE,COST&X3
        BE    BL
        MA    FOUR,X1
        MA    FOUR,X3
        S     ONE,MI
        B     BK
BL      S     MIN,SEND&X3
        MLC   X1,X3
        MLC   W,X2
        MLC   N,NJ
BM      S     NINT,COST&X3
        C     ZERO4,COST&X3
        A     NINTO,COST&X3
        S     ONE,COST&X3
        BE    BJ
        MA    FOUR,X2
        MA    FOURM,X3
        S     ONE,NJ
        B     BM
BN      S     MIN,COL&X2
        MLC   W,X1
        MLC   W,X2
        MLC   W,X3
        MLC   M,MI
        MLC   N,NJ
BO      BWZ   BOO,COL&X2,K
        C     ZERO4,COL&X2
        BH    BP
BOO     MA    FOUR,X2
        MA    FOURM,X3
        S     ONE,NJ
        C     NJ,ZERO4
        BL    BO
        BE    ENDINS
BP      MLC   X1,X3
        MLC   W,X2
        MLC   N,NJ
BQ      BWZ   BQQ,ROW&X1,K
        B     BT
BQQ     MLC   ZERO4,ROW&X1
```

```
265   BR     S      AYTT,COST&X3
266          C      ZERO4,COST&X3
267          BE     BS
268          A      AYTTO,COST&X3
269          S      ONE,COST&X3
270   BS     MA     FOUR,X2
271          MA     FOURM,X3
272          S      ONE,NJ
273          C      NJ,ZERO4
274          BL     BR
275          MA     FOUR,X1
276          MLC    X1,X3
277          MLC    W,X2
278          MLC    N,NJ
279          B      BV
280   BT     MA     FOUR,X1
281          MA     FOUR,X3
282   BV     S      ONE,MI
283          C      MI,ZERO4
284          BL     BQ
285          MLC    W,X1
286          MLC    W,X2
287          MLC    W,X3
288          MLC    M,MI
289   BW     BWZ    BZ,COL&X2,K
290   BX     S      NINT,COST&X3
291          C      ZERO4,COST&X3
292          BE     BY
293          A      NINTO,COST&X3
294          S      ONE,COST&X3
295   BY     MA     FOUR,X1
296          MA     FOUR,X3
297          S      ONE,MI
298          C      MI,ZERO4
299          BL     BX
300          MLC    W,X1
301          MLC    M,MI
302          MA     FOUR,X2
303          B      BZZ
304   BZ     MA     FOUR,X2
305          MA     FOURM,X3
306   BZZ    S      ONE,NJ
307          C      NJ,ZERO4
308          BL     BW
309          B      AE
310   CA     MLC    K,MIN
311          MLC    W,X1
312          MLC    W,X2
313          MLC    W,X3
314          MLC    M,MI
315          MLC    N,NJ
316   CB     BWZ    CE,COL&X2,K
317   CC     BWZ    CD,ROW&X1,K
```

```
                C       MIN,COST&X3
                BH      CD
                MLC     COST&X3,MIN
        CD      MA      FOUR,X1
                MA      FOUR,X3
                S       ONE,MI
                C       MI,ZERO4
                BL      CC
                MLC     W,X1
                MLC     M,MI
                MA      FOUR,X2
                B       CF
        CE      MA      FOUR,X2
                MA      FOURM,X3
        CF      S       ONE,NJ
                C       NJ,ZERO4
                BL      CB
                MLC     W,X1
                MLC     W,X2
                MLC     W,X3
                MLC     M,MI
                MLC     N,NJ
        CG      BWZ     CH,ROW&X1,K
                B       CJ
        CH      A       MIN,COST&X3
                MA      FOUR,X2
                MA      FOURM,X3
                S       ONE,NJ
                C       NJ,ZERO4
                BL      CH
                MLC     W,X2
                MLC     N,NJ
                MA      FOUR,X1
                MLC     X1,X3
                B       CK
        CJ      MA      FOUR,X1
                MA      FOUR,X3
        CK      S       ONE,MI
                C       MI,ZERO4
                BL      CG
                MLC     W,X1
                MLC     W,X2
                MLC     W,X3
                MLC     M,MI
                MLC     N,NJ
        DA      BWZ     DC,COL&X2,K
        DB      S       MIN,COST&X3
                MA      FOUR,X1
                MA      FOUR,X3
                S       ONE,MI
                C       MI,ZERO4
                BL      DB
                MLC     W,X1
```

```
                MLC     M,MI
                MA      FOUR,X2
                B       DD
        DC      MA      FOUR,X2
                MA      FOURM,X3
        DD      S       ONE,NJ
                C       NJ,ZERO4
                BL      DA
                BE      AH
        ENDINS  CS      332
                CS      299
                CC      1
                MLC     @COST@,264
                W
                CC      J
                CS      299
                MLC     M,MI
                MLC     W,X1
                MLC     W,X3
        PCT     MLC     COST&X3,229
                MA      FOURM,X3
                MLC     COST&X3,239
                MA      FOURM,X3
                MLC     COST&X3,249
                MA      FOURM,X3
                MLC     COST&X3,259
                MA      FOURM,X3
                MLC     COST&X3,269
                MA      FOURM,X3
                MLC     COST&X3,279
                MA      FOURM,X3
                MLC     COST&X3,289
                MA      FOURM,X3
                MLC     COST&X3,299
                W
                BCV     OFL1
                B       PD
        OFL1    CC      1
        PD      MA      FOUR,X1
                MLC     X1,X3
                S       ONE,MI
                C       MI,ZERO4
                BL      PCT
                CC      L
                CS      332
                CS      299
                MLC     @SEND@,264
                W
                CC      J
                BCV     OFL2
                B       PSD
        OFL2    CC      1
        PSD     CS      299
```

```
            MLC    M,MI
            MLC    W,X1
            MLC    W,X3
      PG    MLC    SEND&X3,229
            MA     FOURM,X3
            MLC    SEND&X3,239
            MA     FOURM,X3
            MLC    SEND&X3,249
            MA     FOURM,X3
            MLC    SEND&X3,259
            MA     FOURM,X3
            MLC    SEND&X3,269
            MA     FOURM,X3
            MLC    SEND&X3,279
            MA     FOURM,X3
            MLC    SEND&X3,289
            MA     FOURM,X3
            MLC    SEND&X3,299
            W
            BCV    OFL3
            B      PH
      OFL3  CC     1
      PH    MA     FOUR,X1
            MLC    X1,X3
            S      ONE,MI
            C      MI,ZERO4
            BL     PG
            CC     L
            CS     332
            CS     299
            MLC    @ROW@,268
            W
            CC     J
            BCV    OFL4
            B      PK
      OFL4  CC     1
      PK    CS     299
            MLC    W,X1
            MLC    ROW&X1,234
            MA     FOUR,X1
            MLC    ROW&X1,244
            MA     FOUR,X1
            MLC    ROW&X1,254
            MA     FOUR,X1
            MLC    ROW&X1,264
            MA     FOUR,X1
            MLC    ROW&X1,274
            MA     FOUR,X1
            MLC    ROW&X1,284
            MA     FOUR,X1
            MLC    ROW&X1,294
            W
            CC     L
```

```
                BCV     OFL5
                B       PL
        OFL5    CC      1
        PL      CS      332
                CS      299
                MLC     @COLUMN@,265
                W
                CC      J
                BCV     OFL6
                B       PCL
        OFL6    CC      1
        PCL     CS      299
                MLC     W,X2
                MLC     COL&X2,229
                MA      FOUR,X2
                MLC     COL&X2,239
                MA      FOUR,X2
                MLC     COL&X2,249
                MA      FOUR,X2
                MLC     COL&X2,259
                MA      FOUR,X2
                MLC     COL&X2,269
                MA      FOUR,X2
                MLC     COL&X2,279
                MA      FOUR,X2
                MLC     COL&X2,289
                MA      FOUR,X2
                MLC     COL&X2,299
                W
                BLC     STOP
                B       OBJECT
        STOP    H
                END     OBJECT
```